# $Id: PKGBUILD 83148 2013-01-27 16:16:24Z pierre $
# Contributor: Andrea Scarpino <andrea@archlinux.org>
# Contributor: Tobias Kieslich <tobias@archlinux.org>

pkgname=gmpc
pkgver=11.8.16
pkgrel=1
pkgdesc="A GTK2 client for MPD"
arch=('i686' 'x86_64')
url="http://gmpclient.org/home"
license=('GPL')
depends=('libunique' 'libmpd' 'libsoup' 'sqlite' 'hicolor-icon-theme' 'xdg-utils')
makedepends=('gob2' 'intltool' 'vala')
options=('!libtool')
install=gmpc.install
source=(http://download.sarine.nl/Programs/${pkgname}/${pkgver%.*}/${pkgname}-${pkgver}.tar.gz)
md5sums=('223aeb000e41697d8fdf54ccedee89d5')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  sed -i 's|GNOME;AudioVideo;|GTK;AudioVideo;Player;|' data/gmpc.desktop.in
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
